Integration Using the Trapezoidal Rule - Convergence

2004 Autar Kaw, Loubna Guennoun, University of South Florida, kaw@eng.usf.edu, http://numericalmethods.eng.usf.edu

NOTE: This worksheet demonstrates the use of Maple to illustrate the convergence of the trapezoidal rule of integration.

Introduction

Trapezoidal rule is based on the Newton-Cotes formula that if one approximates the integrand of the integral by an nth order polynomial, then the integral of the function is approximated by the integral of that nth order polynomial. Integration of polynomials is simple and is based on the calculus. Trapezoidal rule is the area under the curve for a first order polynomial (straight line). [click here for textbook notes] [click here for power point presentation].

```
> restart;
```

Section I: Input Data

The following simulation illustrates the convergence of Trapezoidal rule of integration. This section is the only section where the user interacts with the program. The user enters any function f(x) and the lower and upper limit of the integration. By entering this data, the program will calculate the exact (Maple numerical value if it is not exact) value of the integral, followed by the results using the Trapezoidal rule *1*, *2*, *3*,...,*n* segments. The program will also display the true error, the absolute relative percentage true error, the approximate error, the absolute relative aprroximate percentage error, and the least number of significant digits correct all as a function of number of segments.

```
> restart;
```

Integrand *f(x)*

```
> f:=x->300*x/(1+exp(x));
```

$$f := x \to \frac{300\,x}{1+e^x}$$

The lower limit of the integral *a*

```
> a:=0.0;
```

$$a := 0.$$

The upper limit of the integral *b*

```
> b:=10.0;
```

$$b := 10.0$$

Maximum number of segments, n

```
> n:=40;
```

$$n := 40$$

This is the end of the user's section. All information must be entered before proceeding to the next section. Re-execute the program.

Section II: Procedure

The following procedure estimates integrals with *n*-segment Trapezoidal rule.
n = number of segments
a = lower limit of integration
b = upper limit of integration
f = integrand

```
> trap:=proc(n,a,b,f)
  local AV,sum,h,i:
  h:=(b-a)/n:
  sum:=0:
  for i from 1 by 1 to n-1 do
  sum:=sum+f(a+i*h):
  end do:
  AV:=(h/2)*(f(a)+2*sum+f(b)):
  return (AV):
  end proc:
```

Section III: Calculation

```
> plot(f(x),x=a..b,title="f(x) vs x",thickness=3, color=black);
```



The exact value of the integral (EV) :

```
> EV:=int(f(x),x=a..b);
```

$EV := 246.5902935$

This loop here calculates the following

AV = approximate value of the integral using the Trapezoidal rule by calling the trap procedure

Et = true error

abs_et = absolute relative true error

Ea = approximate error

ea = absolute relative approximate error

sig = least number of significant digits correct in an approximation

```
> for i from 1 by 1 to n do
  AV[i]:=trap(i,a,b,f):
  Et[i]:=EV-AV[i]:
  abs_et[i]:=abs(Et[i]/EV)*100.0:
  if (i>1) then
  Ea[i]:=AV[i]-AV[i-1]:
  ea[i]:=abs(Ea[i]/AV[i])*100.0:
  sig[i]:=floor((2-log10(ea[i]/0.5))):
  if sig[i]<0 then
  sig[i]:=0:
  end if:
  end if:
  end do:
```

Section IV: Spreadsheet

This table shows the approximate value, true error, absolute relative true error, approximate error and relative approximate error as a function of the number of segments.

```
> with( Spread
  ):EvaluateSpreadsheet(Trapezoidal_Rule_Convergence):
```

	A	B	C	D
1	*The number of segments*	*Exact Value*	*Approximate Value*	*True Error*
2	1	246.5903	0.6810	245.9093
3	2	246.5903	50.5369	196.0534
4	3	246.5903	123.5178	123.0725
5	4	246.5903	170.6119	75.9784
6	5	246.5903	196.8577	49.7326
7	6	246.5903	211.8832	34.7071
8	7	246.5903	221.0657	25.5246
9	8	246.5903	227.0442	19.5461
10	9	246.5903	231.1459	15.4444
11	10	246.5903	234.0802	12.5101
12	11	246.5903	236.2514	10.3389
13	12	246.5903	237.9027	8.6876
14	13	246.5903	239.1878	7.4024
15	14	246.5903	240.2076	6.3827

NOTE: To evaluate the spreadsheet, you need to right click on it and select evaluate

Section V:Graphs

The following graphs shows the approximate value, true error, absolute relative true error, approximate error and relative approximate error as a function of the number of segments

```
> with(plots):
Warning, the name changecoords has been redefined
```

```
> data:=[seq([i,AV[i]],i=1..n)]:
```

```
> pointplot(data,connect=true,color=red,axes=boxed,title="Approxi
  mate value of the integral as a function of number of
  segments",axes=BOXED,labels=["Number of segments","Approximate
  Value"],thickness=3);

> data:=[seq([i,Et[i]],i=1..n)]:
> pointplot(data,connect=true,color=blue,axes=boxed,title="True
  error as a function of number of
  segments",axes=BOXED,labels=["Number of segments","True
  Error"],thickness=3);

> data:=[seq([i,abs_et[i]],i=1..n)]:
> pointplot(data,connect=true,color=blue,axes=boxed,title="Absolu
  te relative true percentage error as a function of number of
  segments",axes=BOXED,labels=["Number of segments","Absolute
  Relative True Error"],thickness=3);

> data:=[seq([i,Ea[i]],i=2..n)]:
> pointplot(data,connect=true,color=green,axes=boxed,title="Appro
  ximate error as a function of number of
  segments",axes=BOXED,labels=["Number of segments","Approximate
  Error"],thickness=3);

> data:=[seq([i,ea[i]],i=2..n)]:
> pointplot(data,connect=true,color=green,axes=boxed,title="Absol
  ute approximate relative percentage error as a function of
  number of segments",axes=BOXED,labels=["Number of segments",
  "Absolute Relative Approximate Error"],thickness=3);

> data:=[seq([i,sig[i]],i=2..n)]:
> pointplot(data,connect=true,color=brown,axes=boxed,title="The
  least correct significant digits as a function of number of
  segments",axes=BOXED,labels=["Number of segments","Least number
  of significant digits"],thickness=3);
```


Approximate value of the integral as a function of number of segments
Approximate Value
250
200
150
100
50
0
10
20
30
40
Number of segments


True error as a function of number of segments
True Error
250
200
150
100
50
0
10
20
30
40
Number of segments

Absolute relative true percentage error as a function of number of segments



Approximate error as a function of number of segments



Absolute approximate relative percentage error as a function of number of segments



The least correct significant digits as a function of number of segments



>

Maple helped us to understand the concept of convergence of multiple segment Trapezoidal rule.

Question:

1. An integral like $\frac{1}{\sqrt{x}}$ from x=0 to 2 cannot be found using Trapezoidal rule as the integrand is infinite at x=0. What if you defined the function as finite at x=0? Would you get close to an accurate solution? Is defining the function at one point x=0 as finite wrong?

References

[1] Autar Kaw, Michael Keteltas, Holistic Numerical Methods Institute, See http://numericalmethods.eng.usf.edu/mws/gen/07int/mws_gen_int_txt_trapcontinuous.doc